// File: @openzeppelin/contracts/access/Ownable.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (access/Ownable.sol)

pragma solidity ^0.8.0;

import "../utils/Context.sol";

/**
 * @dev Contract module which provides a basic access control mechanism, where
 * there is an account (an owner) that can be granted exclusive access to
 * specific functions.
 *
 * By default, the owner account will be the one that deploys the contract. This
 * can later be changed with {transferOwnership}.
 *
 * This module is used through inheritance. It will make available the modifier
 * `onlyOwner`, which can be applied to your functions to restrict their use to
 * the owner.
 */
abstract contract Ownable is Context {
    address private _owner;

    event OwnershipTransferred(address indexed previousOwner, address indexed newOwner);

    /**
     * @dev Initializes the contract setting the deployer as the initial owner.
     */
    constructor() {
        _transferOwnership(_msgSender());
    }

    /**
     * @dev Throws if called by any account other than the owner.
     */
    modifier onlyOwner() {
        _checkOwner();
        _;
    }

    /**
     * @dev Returns the address of the current owner.
     */
    function owner() public view virtual returns (address) {
        return _owner;
    }

    /**
     * @dev Throws if the sender is not the owner.
     */
    function _checkOwner() internal view virtual {
        require(owner() == _msgSender(), "Ownable: caller is not the owner");
    }

    /**
     * @dev Leaves the contract without owner. It will not be possible to call
     * `onlyOwner` functions. Can only be called by the current owner.
     *
     * NOTE: Renouncing ownership will leave the contract without an owner,
     * thereby disabling any functionality that is only available to the owner.
     */
    function renounceOwnership() public virtual onlyOwner {
        _transferOwnership(address(0));
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Can only be called by the current owner.
     */
    function transferOwnership(address newOwner) public virtual onlyOwner {
        require(newOwner != address(0), "Ownable: new owner is the zero address");
        _transferOwnership(newOwner);
    }

    /**
     * @dev Transfers ownership of the contract to a new account (`newOwner`).
     * Internal function without access restriction.
     */
    function _transferOwnership(address newOwner) internal virtual {
        address oldOwner = _owner;
        _owner = newOwner;
        emit OwnershipTransferred(oldOwner, newOwner);
    }
}

// File: @openzeppelin/contracts/security/ReentrancyGuard.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (security/ReentrancyGuard.sol)

pragma solidity ^0.8.0;

/**
 * @dev Contract module that helps prevent reentrant calls to a function.
 *
 * Inheriting from `ReentrancyGuard` will make the {nonReentrant} modifier
 * available, which can be applied to functions to make sure there are no nested
 * (reentrant) calls to them.
 *
 * Note that because there is a single `nonReentrant` guard, functions marked as
 * `nonReentrant` may not call one another. This can be worked around by making
 * those functions `private`, and then adding `external` `nonReentrant` entry
 * points to them.
 *
 * TIP: If you would like to learn more about reentrancy and alternative ways
 * to protect against it, check out our blog post
 * https://blog.openzeppelin.com/reentrancy-after-istanbul/[Reentrancy After Istanbul].
 */
abstract contract ReentrancyGuard {
    // Booleans are more expensive than uint256 or any type that takes up a full
    // word because each write operation emits an extra SLOAD to first read the
    // slot's contents, replace the bits taken up by the boolean, and then write
    // back. This is the compiler's defense against contract upgrades and
    // pointer aliasing, and it cannot be disabled.

    // The values being non-zero value makes deployment a bit more expensive,
    // but in exchange the refund on every call to nonReentrant will be lower in
    // amount. Since refunds are capped to a percentage of the total
    // transaction's gas, it is best to keep them low in cases like this one, to
    // increase the likelihood of the full refund coming into effect.
    uint256 private constant _NOT_ENTERED = 1;
    uint256 private constant _ENTERED = 2;

    uint256 private _status;

    constructor() {
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Prevents a contract from calling itself, directly or indirectly.
     * Calling a `nonReentrant` function from another `nonReentrant`
     * function is not supported. It is possible to prevent this from happening
     * by making the `nonReentrant` function external, and making it call a
     * `private` function that does the actual work.
     */
    modifier nonReentrant() {
        _nonReentrantBefore();
        _;
        _nonReentrantAfter();
    }

    function _nonReentrantBefore() private {
        // On the first call to nonReentrant, _status will be _NOT_ENTERED
        require(_status != _ENTERED, "ReentrancyGuard: reentrant call");

        // Any calls to nonReentrant after this point will fail
        _status = _ENTERED;
    }

    function _nonReentrantAfter() private {
        // By storing the original value once again, a refund is triggered (see
        // https://eips.ethereum.org/EIPS/eip-2200)
        _status = _NOT_ENTERED;
    }

    /**
     * @dev Returns true if the reentrancy guard is currently set to "entered", which indicates there is a
     * `nonReentrant` function in the call stack.
     */
    function _reentrancyGuardEntered() internal view returns (bool) {
        return _status == _ENTERED;
    }
}

// File: @openzeppelin/contracts/token/ERC20/IERC20.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.0) (token/ERC20/IERC20.sol)

pragma solidity ^0.8.0;

/**
 * @dev Interface of the ERC20 standard as defined in the EIP.
 */
interface IERC20 {
    /**
     * @dev Emitted when `value` tokens are moved from one account (`from`) to
     * another (`to`).
     *
     * Note that `value` may be zero.
     */
    event Transfer(address indexed from, address indexed to, uint256 value);

    /**
     * @dev Emitted when the allowance of a `spender` for an `owner` is set by
     * a call to {approve}. `value` is the new allowance.
     */
    event Approval(address indexed owner, address indexed spender, uint256 value);

    /**
     * @dev Returns the amount of tokens in existence.
     */
    function totalSupply() external view returns (uint256);

    /**
     * @dev Returns the amount of tokens owned by `account`.
     */
    function balanceOf(address account) external view returns (uint256);

    /**
     * @dev Moves `amount` tokens from the caller's account to `to`.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transfer(address to, uint256 amount) external returns (bool);

    /**
     * @dev Returns the remaining number of tokens that `spender` will be
     * allowed to spend on behalf of `owner` through {transferFrom}. This is
     * zero by default.
     *
     * This value changes when {approve} or {transferFrom} are called.
     */
    function allowance(address owner, address spender) external view returns (uint256);

    /**
     * @dev Sets `amount` as the allowance of `spender` over the caller's tokens.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * IMPORTANT: Beware that changing an allowance with this method brings the risk
     * that someone may use both the old and the new allowance by unfortunate
     * transaction ordering. One possible solution to mitigate this race
     * condition is to first reduce the spender's allowance to 0 and set the
     * desired value afterwards:
     * https://github.com/ethereum/EIPs/issues/20#issuecomment-263524729
     *
     * Emits an {Approval} event.
     */
    function approve(address spender, uint256 amount) external returns (bool);

    /**
     * @dev Moves `amount` tokens from `from` to `to` using the
     * allowance mechanism. `amount` is then deducted from the caller's
     * allowance.
     *
     * Returns a boolean value indicating whether the operation succeeded.
     *
     * Emits a {Transfer} event.
     */
    function transferFrom(address from, address to, uint256 amount) external returns (bool);
}

// File: @openzeppelin/contracts/utils/Context.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts v4.4.1 (utils/Context.sol)

pragma solidity ^0.8.0;

/**
 * @dev Provides information about the current execution context, including the
 * sender of the transaction and its data. While these are generally available
 * via msg.sender and msg.data, they should not be accessed in such a direct
 * manner, since when dealing with meta-transactions the account sending and
 * paying for execution may not be the actual sender (as far as an application
 * is concerned).
 *
 * This contract is only required for intermediate, library-like contracts.
 */
abstract contract Context {
    function _msgSender() internal view virtual returns (address) {
        return msg.sender;
    }

    function _msgData() internal view virtual returns (bytes calldata) {
        return msg.data;
    }
}

// File: @openzeppelin/contracts/utils/cryptography/MerkleProof.sol



// SPDX-License-Identifier: MIT
// OpenZeppelin Contracts (last updated v4.9.2) (utils/cryptography/MerkleProof.sol)

pragma solidity ^0.8.0;

/**
 * @dev These functions deal with verification of Merkle Tree proofs.
 *
 * The tree and the proofs can be generated using our
 * https://github.com/OpenZeppelin/merkle-tree[JavaScript library].
 * You will find a quickstart guide in the readme.
 *
 * WARNING: You should avoid using leaf values that are 64 bytes long prior to
 * hashing, or use a hash function other than keccak256 for hashing leaves.
 * This is because the concatenation of a sorted pair of internal nodes in
 * the merkle tree could be reinterpreted as a leaf value.
 * OpenZeppelin's JavaScript library generates merkle trees that are safe
 * against this attack out of the box.
 */
library MerkleProof {
    /**
     * @dev Returns true if a `leaf` can be proved to be a part of a Merkle tree
     * defined by `root`. For this, a `proof` must be provided, containing
     * sibling hashes on the branch from the leaf to the root of the tree. Each
     * pair of leaves and each pair of pre-images are assumed to be sorted.
     */
    function verify(bytes32[] memory proof, bytes32 root, bytes32 leaf) internal pure returns (bool) {
        return processProof(proof, leaf) == root;
    }

    /**
     * @dev Calldata version of {verify}
     *
     * _Available since v4.7._
     */
    function verifyCalldata(bytes32[] calldata proof, bytes32 root, bytes32 leaf) internal pure returns (bool) {
        return processProofCalldata(proof, leaf) == root;
    }

    /**
     * @dev Returns the rebuilt hash obtained by traversing a Merkle tree up
     * from `leaf` using `proof`. A `proof` is valid if and only if the rebuilt
     * hash matches the root of the tree. When processing the proof, the pairs
     * of leafs & pre-images are assumed to be sorted.
     *
     * _Available since v4.4._
     */
    function processProof(bytes32[] memory proof, bytes32 leaf) internal pure returns (bytes32) {
        bytes32 computedHash = leaf;
        for (uint256 i = 0; i < proof.length; i++) {
            computedHash = _hashPair(computedHash, proof[i]);
        }
        return computedHash;
    }

    /**
     * @dev Calldata version of {processProof}
     *
     * _Available since v4.7._
     */
    function processProofCalldata(bytes32[] calldata proof, bytes32 leaf) internal pure returns (bytes32) {
        bytes32 computedHash = leaf;
        for (uint256 i = 0; i < proof.length; i++) {
            computedHash = _hashPair(computedHash, proof[i]);
        }
        return computedHash;
    }

    /**
     * @dev Returns true if the `leaves` can be simultaneously proven to be a part of a merkle tree defined by
     * `root`, according to `proof` and `proofFlags` as described in {processMultiProof}.
     *
     * CAUTION: Not all merkle trees admit multiproofs. See {processMultiProof} for details.
     *
     * _Available since v4.7._
     */
    function multiProofVerify(
        bytes32[] memory proof,
        bool[] memory proofFlags,
        bytes32 root,
        bytes32[] memory leaves
    ) internal pure returns (bool) {
        return processMultiProof(proof, proofFlags, leaves) == root;
    }

    /**
     * @dev Calldata version of {multiProofVerify}
     *
     * CAUTION: Not all merkle trees admit multiproofs. See {processMultiProof} for details.
     *
     * _Available since v4.7._
     */
    function multiProofVerifyCalldata(
        bytes32[] calldata proof,
        bool[] calldata proofFlags,
        bytes32 root,
        bytes32[] memory leaves
    ) internal pure returns (bool) {
        return processMultiProofCalldata(proof, proofFlags, leaves) == root;
    }

    /**
     * @dev Returns the root of a tree reconstructed from `leaves` and sibling nodes in `proof`. The reconstruction
     * proceeds by incrementally reconstructing all inner nodes by combining a leaf/inner node with either another
     * leaf/inner node or a proof sibling node, depending on whether each `proofFlags` item is true or false
     * respectively.
     *
     * CAUTION: Not all merkle trees admit multiproofs. To use multiproofs, it is sufficient to ensure that: 1) the tree
     * is complete (but not necessarily perfect), 2) the leaves to be proven are in the opposite order they are in the
     * tree (i.e., as seen from right to left starting at the deepest layer and continuing at the next layer).
     *
     * _Available since v4.7._
     */
    function processMultiProof(
        bytes32[] memory proof,
        bool[] memory proofFlags,
        bytes32[] memory leaves
    ) internal pure returns (bytes32 merkleRoot) {
        // This function rebuilds the root hash by traversing the tree up from the leaves. The root is rebuilt by
        // consuming and producing values on a queue. The queue starts with the `leaves` array, then goes onto the
        // `hashes` array. At the end of the process, the last hash in the `hashes` array should contain the root of
        // the merkle tree.
        uint256 leavesLen = leaves.length;
        uint256 proofLen = proof.length;
        uint256 totalHashes = proofFlags.length;

        // Check proof validity.
        require(leavesLen + proofLen - 1 == totalHashes, "MerkleProof: invalid multiproof");

        // The xxxPos values are "pointers" to the next value to consume in each array. All accesses are done using
        // `xxx[xxxPos++]`, which return the current value and increment the pointer, thus mimicking a queue's "pop".
        bytes32[] memory hashes = new bytes32[](totalHashes);
        uint256 leafPos = 0;
        uint256 hashPos = 0;
        uint256 proofPos = 0;
        // At each step, we compute the next hash using two values:
        // - a value from the "main queue". If not all leaves have been consumed, we get the next leaf, otherwise we
        //   get the next hash.
        // - depending on the flag, either another value from the "main queue" (merging branches) or an element from the
        //   `proof` array.
        for (uint256 i = 0; i < totalHashes; i++) {
            bytes32 a = leafPos < leavesLen ? leaves[leafPos++] : hashes[hashPos++];
            bytes32 b = proofFlags[i]
                ? (leafPos < leavesLen ? leaves[leafPos++] : hashes[hashPos++])
                : proof[proofPos++];
            hashes[i] = _hashPair(a, b);
        }

        if (totalHashes > 0) {
            require(proofPos == proofLen, "MerkleProof: invalid multiproof");
            unchecked {
                return hashes[totalHashes - 1];
            }
        } else if (leavesLen > 0) {
            return leaves[0];
        } else {
            return proof[0];
        }
    }

    /**
     * @dev Calldata version of {processMultiProof}.
     *
     * CAUTION: Not all merkle trees admit multiproofs. See {processMultiProof} for details.
     *
     * _Available since v4.7._
     */
    function processMultiProofCalldata(
        bytes32[] calldata proof,
        bool[] calldata proofFlags,
        bytes32[] memory leaves
    ) internal pure returns (bytes32 merkleRoot) {
        // This function rebuilds the root hash by traversing the tree up from the leaves. The root is rebuilt by
        // consuming and producing values on a queue. The queue starts with the `leaves` array, then goes onto the
        // `hashes` array. At the end of the process, the last hash in the `hashes` array should contain the root of
        // the merkle tree.
        uint256 leavesLen = leaves.length;
        uint256 proofLen = proof.length;
        uint256 totalHashes = proofFlags.length;

        // Check proof validity.
        require(leavesLen + proofLen - 1 == totalHashes, "MerkleProof: invalid multiproof");

        // The xxxPos values are "pointers" to the next value to consume in each array. All accesses are done using
        // `xxx[xxxPos++]`, which return the current value and increment the pointer, thus mimicking a queue's "pop".
        bytes32[] memory hashes = new bytes32[](totalHashes);
        uint256 leafPos = 0;
        uint256 hashPos = 0;
        uint256 proofPos = 0;
        // At each step, we compute the next hash using two values:
        // - a value from the "main queue". If not all leaves have been consumed, we get the next leaf, otherwise we
        //   get the next hash.
        // - depending on the flag, either another value from the "main queue" (merging branches) or an element from the
        //   `proof` array.
        for (uint256 i = 0; i < totalHashes; i++) {
            bytes32 a = leafPos < leavesLen ? leaves[leafPos++] : hashes[hashPos++];
            bytes32 b = proofFlags[i]
                ? (leafPos < leavesLen ? leaves[leafPos++] : hashes[hashPos++])
                : proof[proofPos++];
            hashes[i] = _hashPair(a, b);
        }

        if (totalHashes > 0) {
            require(proofPos == proofLen, "MerkleProof: invalid multiproof");
            unchecked {
                return hashes[totalHashes - 1];
            }
        } else if (leavesLen > 0) {
            return leaves[0];
        } else {
            return proof[0];
        }
    }

    function _hashPair(bytes32 a, bytes32 b) private pure returns (bytes32) {
        return a < b ? _efficientHash(a, b) : _efficientHash(b, a);
    }

    function _efficientHash(bytes32 a, bytes32 b) private pure returns (bytes32 value) {
        /// @solidity memory-safe-assembly
        assembly {
            mstore(0x00, a)
            mstore(0x20, b)
            value := keccak256(0x00, 0x40)
        }
    }
}

// File: contracts/crypto/AllowedEvaluator.sol



// SPDX-License-Identifier: CC0
// Copyright (c) 2022 unReal Accelerator, LLC (https://unrealaccelerator.io)
pragma solidity ^0.8.19;

/// @author: jason@unrealaccelerator.io
/// @title: AllowedEvaluator
/// @notice Merkle Proof for allow list implementations with amount

import "@openzeppelin/contracts/utils/cryptography/MerkleProof.sol";

contract AllowedEvaluator {
    bytes32 internal _allowedMerkleRoot;

    function _setAllowedMerkleRoot(bytes32 allowedMerkleRoot_) internal {
        _allowedMerkleRoot = allowedMerkleRoot_;
    }

    function _validateMerkleProof(
        uint256 index,
        address account,
        uint256 amount,
        bytes32[] calldata merkleProof
    ) internal view returns (bool) {
        return
            MerkleProof.verify(
                merkleProof,
                _allowedMerkleRoot,
                keccak256(abi.encodePacked(index, account, amount))
            );
    }
}

// File: contracts/interfaces/IStakingForClaims.sol



// SPDX-License-Identifier: CC0
// Copyright (c) 2022 unReal Accelerator, LLC (https://unrealaccelerator.io)
pragma solidity ^0.8.9;

/// @title IStakingForClaims
/// @author jason@unrealaccelerator.io
/// @notice Interface for getting count of claims

interface IStakingForClaims {
    function balanceOf(address account) external view returns (uint256);
}

// File: contracts/ShillingTokenClaimsV2.sol



// SPDX-License-Identifier: CC0
// Copyright (c) 2022 unReal Accelerator, LLC (https://unrealaccelerator.io)
pragma solidity ^0.8.19;

/**
 * @title ShillingTokenClaimsV2
 * @dev A contract for managing claims of Shilling tokens from staking contracts.
 * @author jason@unrealaccelerator.io
 */

import "@openzeppelin/contracts/security/ReentrancyGuard.sol";
import "@openzeppelin/contracts/access/Ownable.sol";
import "@openzeppelin/contracts/token/ERC20/IERC20.sol";
import "./crypto/AllowedEvaluator.sol";
import "./interfaces/IStakingForClaims.sol";

contract ShillingTokenClaimsV2 is ReentrancyGuard, Ownable, AllowedEvaluator {
    IERC20 public tokenContract; // ERC20 token contract for Shilling tokens
    IStakingForClaims public stakingCOKContract; // Staking contract for COK tokens
    IStakingForClaims public stakingHCOKContract; // Staking contract for HCOK tokens

    uint256 public ethPerThreeShillings = 0.22 ether; // Default eth amount for 3 shilling

    // mapping to track claimed amounts per address
    mapping(address => bool) public addressClaimed;

    ///////// Security

    address public administrator; // Address with administrative privileges
    bool public claimAllowed; // Used to control claim events
    bool public purchaseAllowed; // Used to control shilling purchase events

    ///////// Errors

    error NotAuthorized(); // Error thrown when caller is not authorized
    error NoBalanceToWithdraw(); // Error thrown when attempting to withdraw with no balance
    error ClaimNotAllowed(); // Error thrown with attempting to claim but claiming is not allowed
    error ClaimNotValid(); // Error thrown with attempting to claim but claiming is not allowed
    error InvalidNumberOfTokens();
    error PurchaseNotAllowed();
    error AddressAlreadyClaimed();
    error InsufficientTokens();
    error IncorrectAmountOfETH();

    /////////  Events

    event ClaimAllowed(bool indexed state);
    event PurchaseAllowed(bool indexed state);
    event Claimed(address indexed user, uint256 amount);
    event ClaimsPurchased(
        address indexed user,
        uint256 amount,
        uint256 totalEth
    );

    ///////// Modifiers

    /**
     * @dev Modifier to check for authorized roles (Owner or Administrator)
     */
    modifier onlyAuthorized() {
        validateAuthorized();
        _;
    }

    function validateAuthorized() private view {
        if (_msgSender() != owner() && _msgSender() != administrator)
            revert NotAuthorized();
    }

    /**
     * @dev Constructor to initialize the contract with required addresses.
     * @param tokenContract_ Address of the Shilling token contract.
     * @param stakingCOKContract_ Address of the staking contract for COK tokens.
     * @param stakingHCOKContract_ Address of the staking contract for HCOK tokens.
     * @param administrator_ Address with administrative privileges.
     */
    constructor(
        address tokenContract_,
        address stakingCOKContract_,
        address stakingHCOKContract_,
        address administrator_
    ) {
        tokenContract = IERC20(tokenContract_);
        stakingCOKContract = IStakingForClaims(stakingCOKContract_);
        stakingHCOKContract = IStakingForClaims(stakingHCOKContract_);
        administrator = administrator_;
    }

    /**
     * @dev Fallback function to receive Ether.
     * This function is payable and allows the contract to receive ETH.
     */
    receive() external payable {}

    /**
     * @dev Get the current supply of Shilling tokens held by the contract.
     * @return The balance of Shilling tokens held by the contract.
     */
    function shillingSupply() external view returns (uint256) {
        return uint256(tokenContract.balanceOf(address(this)));
    }

    /**
     * @dev Calculates the total number of allowed claims.
     * @param account The address of the account to check allowed claims for.
     * @return The number of allowed claims.
     */
    function allowedClaims(address account) public view returns (uint256) {
        return (allowedClaimsWithCOK(account) + allowedClaimsWithHCOK(account));
    }

    /**
     * @dev Calculates the number of allowed claims based on the staked COK balance.
     * @param account The address of the account to check allowed claims for.
     * @return The number of allowed claims. Each staked COK token grants 1 claim, rounded down to the nearest multiple of 3.
     */
    function allowedClaimsWithCOK(
        address account
    ) public view returns (uint256) {
        uint256 balance = stakingCOKContract.balanceOf(account);
        // 1 claim per staked COK; but only in multiples of 3
        uint256 claims = (balance / 3) * 3;
        return claims;
    }

    /**
     * @dev Get the number of allowed claims for an account using the stakingHCOKContract.
     * @param account The address of the account to check allowed claims for.
     * @return The number of allowed claims based on the staked HCOK balance.
     */
    function allowedClaimsWithHCOK(
        address account
    ) public view returns (uint256) {
        return (stakingHCOKContract.balanceOf(account) * 9);
    }

    function claim(
        uint256 index,
        uint256 amount,
        bytes32[] calldata merkleProof
    ) external nonReentrant {
        if (
            !AllowedEvaluator._validateMerkleProof(
                index,
                _msgSender(),
                amount,
                merkleProof
            )
        ) revert ClaimNotValid();

        // Ensure the address has not already claimed
        if (addressClaimed[_msgSender()]) revert AddressAlreadyClaimed();

        // Ensure claiming is allowed
        if (!claimAllowed) revert ClaimNotAllowed();

        // Calculate the balance of the contract's token holdings
        uint256 balance = uint256(tokenContract.balanceOf(address(this)));

        // Calculate the reward to be transferred to the account in wei
        uint256 rewardWei = amount * 10 ** 18;

        // Ensure the contract has sufficient tokens to cover the reward
        if (rewardWei > balance) revert InsufficientTokens();

        // Transfer the reward tokens to the account
        tokenContract.transfer(_msgSender(), rewardWei);

        // Set that the address has made their claims
        addressClaimed[_msgSender()] = true;

        // Emit a Claimed event to indicate a successful claim
        emit Claimed(_msgSender(), amount);
    }

    /**
     * @dev Allows an account to purchase rewards using ETH.
     * The function checks if claiming is allowed, calculates the number of claims that can be purchased,
     * verifies that the number of claims is a multiple of 3, calculates the total ETH required,
     * transfers the ETH to the contract, and emits a `ClaimsPurchased` event.
     * This function is non-reentrant.
     * @notice Only addresses can purchase claims in multiples of 3.
     * @notice Emits a `ClaimsPurchased` event upon successful purchase.
     * @param numberOfShillings The number of claims to purchase (must be a multiple of 3).
     */
    function claimWithEth(
        uint256 numberOfShillings
    ) external payable nonReentrant {
        // Ensure claiming is allowed
        if (!purchaseAllowed) revert PurchaseNotAllowed();

        // Verify that the number of claims is a multiple of 3
        if (numberOfShillings % 3 != 0 || numberOfShillings == 0)
            revert InvalidNumberOfTokens();

        // Calculate the total ETH required for the claims
        uint256 totalEthRequired = (numberOfShillings / 3) *
            ethPerThreeShillings;

        // Ensure the sent ETH matches the required amount
        if (msg.value != totalEthRequired) revert IncorrectAmountOfETH();

        // Calculate the balance of the contract's token holdings
        uint256 balance = uint256(tokenContract.balanceOf(address(this)));

        uint256 numberOfShillingsWei = numberOfShillings * 10 ** 18;

        // Ensure the contract has sufficient tokens to cover the reward
        if (int256(numberOfShillingsWei) - int256(balance) > 0)
            revert InsufficientTokens();

        // Transfer the Shillings to the account
        tokenContract.transfer(_msgSender(), uint256(numberOfShillingsWei));

        // Emit a ClaimsPurchased event to indicate a successful purchase
        emit ClaimsPurchased(
            _msgSender(),
            numberOfShillingsWei,
            totalEthRequired
        );
    }

    /**
     * @dev Sets the merkle root for claiming verification
     * Only authorized addresses can call this function.
     */
    function setMerkleRoot(bytes32 _merkleRoot) external onlyAuthorized {
        AllowedEvaluator._setAllowedMerkleRoot(_merkleRoot);
    }

    /**
     * @dev Gets the merkle root for claiming verification
     * Only authorized addresses can call this function.
     */
    function merkleRoot() external view returns (bytes32) {
        return AllowedEvaluator._allowedMerkleRoot;
    }

    /**
     * @dev Activate/disable claiming
     * Only authorized addresses can call this function.
     */
    function setClaimAllowed(bool claimAllowed_) external onlyAuthorized {
        claimAllowed = claimAllowed_;
        emit ClaimAllowed(claimAllowed);
    }

    /**
     * @dev Activate/disable purchasing
     * Only authorized addresses can call this function.
     */
    function setPurchaseAllowed(bool purchaseAllowed_) external onlyAuthorized {
        purchaseAllowed = purchaseAllowed_;
        emit PurchaseAllowed(purchaseAllowed);
    }

    /**
     * @dev Set/update the token contract address
     * Only authorized addresses can call this function.
     */
    function setTokenContract(address tokenContract_) external onlyAuthorized {
        tokenContract = IERC20(tokenContract_);
    }

    /**
     * @dev Set/update the COK contract address
     * Only authorized addresses can call this function.
     */
    function setStakingCOKContract(
        address stakingCOKContract_
    ) external onlyAuthorized {
        stakingCOKContract = IStakingForClaims(stakingCOKContract_);
    }

    /**
     * @dev Set/update the HCOK contract address
     * Only authorized addresses can call this function.
     */
    function setStakingHCOKContract(
        address stakingHCOKContract_
    ) external onlyAuthorized {
        stakingHCOKContract = IStakingForClaims(stakingHCOKContract_);
    }

    /**
     * @dev Set/update the eth required to claim 3 shillings
     * Amount must be in wei
     * Only authorized addresses can call this function.
     */
    function setEthPerThreeShillings(
        uint256 ethPerThreeShillings_
    ) external onlyAuthorized {
        ethPerThreeShillings = ethPerThreeShillings_;
    }

    /**
     * @dev Admin function to withdraw unclaimed Shilling tokens after the claiming period.
     * Only authorized addresses can call this function.
     */
    function withdrawRemainingToken() external onlyAuthorized {
        uint256 amount = tokenContract.balanceOf(address(this));
        tokenContract.transfer(owner(), amount);
        delete amount;
    }

    /**
     * @dev Admin function to withdraw ETH from the contract.
     * Only authorized addresses can call this function.
     */
    function withdraw() external onlyAuthorized {
        uint256 balance = address(this).balance;
        if (balance <= 0) revert NoBalanceToWithdraw();

        address payable ownerAddress = payable(owner());
        ownerAddress.transfer(balance);
    }
}